Lite86 (old version)

DeCompress:


  ld   a, (hl)

        bit  7, a

        jr   z, Compressed

        inc  hl
        and  01111111b

        ld   b, 0
        ld   c, a

        ldir

        jr   Decompress

Compressed:

        push af

        or   11111100b

        ld   b, a

        inc  hl
        ld   c, (hl)

        inc  hl

        pop  af

        and  01111100b

        rrca

        rrca

        or   a
        ret  z

        push hl

        ld   h, d

        ld   l, e

        add  hl, bc

        inc  a

        inc  a

        ld   b, 0
        ld   c, a

        ldir

        pop  hl

        jr   Decompress

Let’s start by the easy stuff.  Take a look at the 2 green highlited instruction.  2 instructions doing the exact same thing have been literally “copy/pasted”  in the 2 possibilities.  Of course, the obvious one here is to increment hl right after loading a.  Result?  Saving 1 byte.

Now take a look at the routine, see the yellow tags?  Everytime that area is called besides the first time, bc should be 0 since it’ll pass through the “ldir”  so right off the bat, you can kill that yellow line safelly without any drawbacks as long as you make sure b = 0 upon enterring the routine.  So end up saving 2 byte here.

Noticed the purple tag?  This commands is there to check wether the register a is null or not.  Simple.  But not lean.  There’s a bunch of instructions above which return the state of register a once played with.  Notably the 3 upper commands.  So want the solution?  Simply kill it!  So 1 more byte.

Ok so at this point, all the trivial optimizations are done.  Now we have to go and look deeper in the code to snatch the implementation optimization.  They are usually the ones to yield the best results and I think this case is a pretty good one to ilustrate this.  Let’s update the code to simplify things.


  ld   a, (hl)

        inc  hl

        bit  7, a

        jr   z, Compressed

        and  01111111b

        ld   c, a

        ldir

        jr   Decompress

Compressed:

        push af

        or   11111100b

        ld   b, a
        ld   c,(hl)
        inc  hl
        pop  af

        and  01111100b

        rrca

        rrca

        ret  z

        push hl
        ld   h, d
        ld   l, e
        add  hl, bc
        inc  a

        inc  a

        ld   c, a

        ldir

        pop  hl

        jr   Decompress

The next part is much more tricky though.  The basic idea is that we want to remove the “ld  b, 0” so that we don’t touch bc at all  hence we have to use “hl” since “de” must contain the destination address. We’ll also change “inc  hl” so that it appears after the pop.  This way, we can free hl sooner.  Now when it comes down to adding the destination to the offset, it’s a piece of cake, since hl already contains what used to be “bc” we don’t have to add it.  Instead, we can simply add the destination to the offset (hl).  That might’ve sounded tedious, but all in one, this is what it should look like after the implementation.

        push hl

        push af

        or   11111100b

        ld   l,(hl)

        ld   h, a

        pop  af

        and  01111100b
        rrca
        rrca
        ret  z
        add  hl, de

        inc  a
        inc  a
        ld   c, a
        ldir

        pop  hl

        inc  hl

        jr   Decompress

Looks pretty good eh?  Not really.  I’ve seperated this whole thing into 2 steps.  Take a look at the ret in the code.  Now take a look at the 2 pushes and find their matching pop.  You’ll notice there’s a little glitch there.  =)  Not really a glitch because even more optimization will fix that thing.  And remember, we’re all doing this to save 4 nasty little byte in this case =].

The dark blue part is again a tad tricky but not so bad.  The idea is that we wish to cancel out the “and” since that’s 2 bytes.  We know the byte that gets there is < 128 which means the first bit is off and the routine shifts 2 times to the right so technically, we should only move the whole string 2 times right and we wouldn’t even need the push since shifting right can be done directly on a register ; “a” would be free.  So that’s exaclty what we’ll do.  Instead of using the intermediate register a, we’ll use c directly and shift him twice remove all the stack crap so we’ll be able to preserve “a” and therefore “push hl” later in the code fixing the “ret” bug.

Compressed:

        ld   c, a
        sra  c

        sra  c

        ret  z

        inc  c

        inc  c

        push hl

        ld   l,(hl)

        or   11111100b

        ld   h,a

        add  hl,de

        ldir

        pop  hl

        inc  hl

        jr   Decompress

Okey dokey, well this time around, we’re looking at the red tag trying to optimize the bit command.  You’ll notice there’s a “ld  c, a” in the compression function which we can safelly moved up.  We’ll also be moving the decompression’s “and” command up and now instead of checking for the bit to be on or off, we’ll check to see if there’s any diffrence between a and c.  If there is, then it’s because the bit was set.  Sounds good?  1 byte will be killed in the process.  Let’s also convert the “and” in the uncompression routine to a “res”.  This way, we can illiminate the need to reload “c” with the value of the “andded a” by using “res” on “c” directly therefore saving another byte.
Well, it’s 3 :30am on my side of the globe, I’m tired and in exams so I think I’ll stop thinking and dump the results.  Here it is.  Anyhow I don’t think I could compress this code more at this point.  We went from 43 bytes to 31 bytes.  That’s 12bytes less : 28% less.  Which is really good for size optimization.  It just shows what you can do when you sit down and check every single lines of code one by one thinking about every possibilities  (Or just about :o).

Decompress:                                 ; Assumes that b = 0.

        ld   a, (hl)



inc  hl

        ld   c, a



res  7, c

        cp   c

        jr   z, Compressed

        ldir

        jr   Decompress

Compressed:

        sra  c

        ret  z

        sra  c

        inc  c

        inc  c

        push hl

        or   11111100b

        ld   h,a

        ld   l, (hl)

        add  hl, de

        ldir

        pop  hl

        inc  hl

        jr   Decompress

